
Abstract 
 

DSKDIRXP.BAS is presented; a BASIC Language program to output an Extended Disk 
Directory to the Printer.  

 
——  
 
This paper and its associated code are available online at:  
 
  http://www.bds-soft.com/cocoPapers.php . 
 
===== 

  

http://www.bds-soft.com/cocoPapers.php

Introduction 
 

Have you ever been frustrated by a long Disk Directory that scrolled off the screen before 
you could read it all. There are several solutions available for this problem. Perhaps the prettiest 
is the simple: 

 
POKE 111,254:DIR 
 

Which quickly sends the directory from Drive 0 straight to the printer. 
 

This is indeed simple and quick. But I wanted more. I wanted: 
 

1. To be able to send the directory from any one of the four drives to the printer. 
 
2. To have a “full” directory listing, including for each file: a list of all the disk 

granules used in order, the number of sectors used in the last granule, and 
the number of bytes used in the last sector. 

 
3. To have the program also report the number of “Free” granules remaining on 

the disk. 
 
4. To have the fun of writing it myself. 
 

The result is DSKDIRXP.BAS, the Extended Disk Directory To Printer program 
presented herein. It is written entirely in BASIC, and it’s very (VERY) slow compared to the 
above POKE mechanism, but I don’t care if it’s slow, as long as it does all the extra stuff I 
wanted. 

 
——— 

 
In works of even this relative simplicity, typos and other errors are likely to sneak in. 

Please let me know about any you discover so I can note and correct them. 
 
M.D.J. 2021/11/06 
info@bds-soft.com 
 
===== 

DSKDIRXP.BAS 

Extended Disk Directory 

To Printer 
 

Each line of the directory printout (except the last) looks like this: 
 

 
 
Note that the “>” and “>>” symbols in the listing have no purpose other than to 

differentiate between granules, sectors, and bytes. Specifically, “>” does NOT herein indicate 
“Greater Than”. 

 
The final line just looks like: 
 
  FREE =  16 
 
And (Tandy 114) reminds us that the File Types are: 
 

0 = BASIC program 
1 = BASIC data file 
2 = Machine-language program 
3 = Text editor source file 

 
and the ASCII Flag is: 
 

A => &HFF = the file is in ASCII format 
B => 0 = the file is in binary format 

 
———  

The BASIC Language Program: 
 

1000 '***** 
1010 '* 
1020 '* DSKDIRXP.BAS 
1030 '* MDJ 2021/10/17 
1040 '* 
1050 '* EXTENDED 
1060 '* DISK DIRECTORY 
1070 '* TO PRINTER 
1080 '* 
1090 '***** 
1110 ' 
1200 '***** 
1210 '* 
1220 '* INITIAL SETUP 
1230 '* 
1240 '***** 
1250 ' 
1260 PCLEAR 1 
1270 CLEAR 8192 
1280 ' 
1300 '***** 
1310 '* 
1320 '* SPLASH SCREEN 
1330 '* 
1340 '* A$ = USER RESPONSE 
1350 '* 
1360 '***** 
1370 ' 
1380 CLS 
1390 PRINT@ 106, "DSKDIRXP.BAS" 
1400 PRINT@ 140, "EXTENDED" 
1410 PRINT@ 165, "DISK DIRECTORY PRINTER" 
1420 PRINT@ 262, "BY  M. DAVID JOHNSON" 
1430 PRINT@ 299, "2021/10/17" 
1440 PRINT@ 386, "(PRESS ANY KEY  TO CONTINUE)" 
1450 A$=INKEY$ 
1460 IF A$="" GOTO 1450 
1470 ' 
1500 '***** 
1510 '* 
1520 '* SELECT DISK DRIVE 
1530 '* 
1540 '* DN = DISK DRIVE # 
1550 '* 

1560 '***** 
1570 ' 
1580 CLS 
1590 PRINT@ 66, "PLEASE CHOOSE:" 
1600 PRINT@ 130, "(0) DISK DRIVE 0" 
1610 PRINT@ 162, "(1) DISK DRIVE 1" 
1620 PRINT@ 194, "(2) DISK DRIVE 2" 
1630 PRINT@ 226, "(3) DISK DRIVE 3" 
1640 A$=INKEY$ 
1650 IF A$="" GOTO 1640 
1660 IF A$="0" GOTO 1720 
1670 IF A$="1" GOTO 1740 
1680 IF A$="2" GOTO 1760 
1690 IF A$="3" GOTO 1780 
1700 PRINT@ 386, "(PLEASE PRESS 0, 1, 2, OR 3)" 
1710 GOTO 1640 
1720 DN=0 
1730 GOTO 1790 
1740 DN=1 
1750 GOTO 1790 
1760 DN=2 
1770 GOTO 1790 
1780 DN=3 
1790 PRINT@ 290, "DISK DRIVE#";DN 
1800 PRINT@ 322, "HAS BEEN SELECTED" 
1810 PRINT@ 386, "(PRESS ANY KEY TO CONTINUE)" 
1820 A$=INKEY$ 
1830 IF A$="" GOTO 1820 
1840 ' 
1850 CLS 
1860 PRINT@ 66, "PROCESSING" 
1870 PRINT@ 130, "(PLEASE WAIT)" 
1880 PRINT@ 194, "*"; 
1890 ' 
1900 '***** 
1910 '* 
1920 '* READ THE FILE 
1930 '* ALLOCATION TABLE 
1940 '* 
1950 '* TN = TRACK NUMBER 
1960 '* SN = SECTOR NUMBER 
1970 '* B$ = FIRST DISK BUFFER 
1980 '* C$ = SECOND DISK BUFFER 
1990 '* FA = FAT ARRAY 
2000 '* T$ = TEMPORARY STRING 
2010 '* 
2020 '***** 

2030 ' 
2040 DIM FA(68), B$(12), C$(12) 
2050 ' 
2060 'GET THE FAT 
2070 TN=17 
2080 SN=2 
2090 PRINT "*"; 
2100 DSKI$ DN, TN, SN, B$(SN), C$(SN) 
2110 ' 
2120 'FORM THE FAT ARRAY 
2130 FOR I=0 TO 67 
2140 T$=MID$(B$(SN),I+1,1) 
2150 FA(I)=ASC(T$) 
2160 ' 
2170 'PROGRESS COUNTING 
2180 IF I=10 THEN PRINT "*"; 
2190 IF I=20 THEN PRINT "*"; 
2200 IF I=30 THEN PRINT "*"; 
2210 IF I=40 THEN PRINT "*"; 
2220 IF I=50 THEN PRINT "*"; 
2230 IF I=60 THEN PRINT "*"; 
2240 NEXT I 
2250 PRINT "*"; 
2260 ' 
2300 '***** 
2310 '* 
2320 '* READ THE 
2330 '* DISK DIRECTORY 
2340 '* 
2350 '***** 
2360 ' 
2370 'STEP THROUGH THE 
2380 'PERTINENT SECTORS 
2390 FOR SN=3 TO 11 
2400 DSKI$ DN, TN, SN, B$(SN), C$(SN) 
2410 ' 
2420 'PROGRESS COUNTING 
2430 PRINT "*"; 
2440 NEXT SN 
2450 PRINT "*"; 
2460 ' 
2500 '***** 
2510 '* 
2520 '* BREAK THE DISK 
2530 '* BUFFERS INTO 
2540 '* RAW DIRECTORY 
2550 '* BUFFERS 

2560 '* 
2570 '* R$ = RAW BUFFER 
2580 '* T1 = TEMP VARIABLE 
2590 '* 
2600 '***** 
2610 ' 
2620 DIM R$(72) 
2630 ' 
2640 'STEP THROUGH THE 
2650 'PERTINENT SECTORS 
2660 T1 = -1 
2670 FOR SN = 3 TO 11 
2680 T1=T1+1 
2690 R$(T1)=MID$(B$(SN), 1, 32) 
2700 T1=T1+1 
2710 R$(T1)=MID$(B$(SN), 33, 32) 
2720 T1=T1+1 
2730 R$(T1)=MID$(B$(SN), 65, 32) 
2740 T1=T1+1 
2750 R$(T1)=MID$(B$(SN), 97, 32) 
2760 T1=T1+1 
2770 R$(T1)=MID$(C$(SN), 1, 32) 
2780 T1=T1+1 
2790 R$(T1)=MID$(C$(SN), 33, 32) 
2800 T1=T1+1 
2810 R$(T1)=MID$(C$(SN), 65, 32) 
2820 T1=T1+1 
2830 R$(T1)=MID$(C$(SN), 97, 32) 
2840 ' 
2850 'PROGRESS COUNTING 
2860 PRINT "*"; 
2870 NEXT SN 
2880 PRINT@ 226, "*"; 
2890 ' 
2900 '***** 
2910 '* 
2920 '* FORM DIRECTORY 
2930 '* LISTING ENTRIES 
2940 '* 
2950 '* F$ = FILENAME 
2960 '* E$ = EXTENSION 
2970 '* G$ = 8.3 FILENAME 
2980 '* FT = FILE TYPE 
2990 '* H$ = ASCII FLAG 
3000 '* FG = FIRST GRANULE # 
3010 '* SB = NUMBER OF BYTES 
3020 '*       IN LAST SECTOR 

3030 '* T2,T3 = TEMP VARS 
3040 '* 
3050 '***** 
3060 ' 
3070 DIM F$(72), E$(72), G$(72) 
3080 DIM FT(72), H$(72) 
3090 DIM FG(72), SB(72) 
3100 ' 
3110 'STEP THROUGH THE 
3120 'RAW BUFFERS 
3130 FOR T1=0 TO 71 
3140 ' 
3150 'FILENAME 
3160 F$(T1)=MID$(R$(T1), 1, 8) 
3170 E$(T1)=MID$(R$(T1), 9, 3) 
3180 G$(T1)=F$(T1)+"."+E$(T1) 
3190 ' 
3200 'FILE TYPE 
3210 T$=MID$(R$(T1), 12, 1) 
3220 FT(T1)=ASC(T$) 
3230 ' 
3240 'ASCII FLAG 
3250 T$=MID$(R$(T1), 13, 1) 
3260 T2=ASC(T$) 
3270 H$(T1)="A" 
3280 IF T2=0 THEN H$(T1)="B" 
3290 ' 
3300 'FIRST GRANULE # 
3310 T$=MID$(R$(T1), 14, 1) 
3320 FG(T1)=ASC(T$) 
3330 ' 
3340 '#BYTES IN LAST SECTOR 
3350 T$=MID$(R$(T1), 15, 1) 
3360 T2=ASC(T$) 
3370 T$=MID$(R$(T1), 16, 1) 
3380 T3=ASC(T$) 
3390 SB(T1)=(T2*256)+T3 
3400 ' 
3410 'PROGRESS COUNTING 
3420 PRINT "*"; 
3430 IF T1=26 THEN PRINT@ 258, "*"; 
3440 IF T1=53 THEN PRINT@ 290, "*"; 
3450 NEXT T1 
3460 PRINT " -- DONE" 
3470 ' 
9100 '***** 
9110 '* 

9120 '* SEND REPORT 
9130 '* TO PRINTER 
9140 '* 
9150 '***** 
9160 ' 
9170 PRINT@ 66, "          " 
9180 PRINT@ 130, "             " 
9190 PRINT@ 98, "READY THE PRINTER" 
9200 PRINT@ 386, "(PRESS ANY KEY TO CONTINUE)"  
9210 A$=INKEY$ 
9220 IF A$="" GOTO 9210 
9230 ' 
9240 'STEP THROUGH THE 
9250 'DIRECTORY ENTRIES 
9260 FOR T1=0 TO 71 
9270 ' 
9280 'CHECK FILENAME  
9290 'FIRST CHARACTER 
9300 'VALIDITY 
9310 T2=ASC(G$(T1)) 
9320 IF T2=0 GOTO 9770 
9330 IF T2=255 GOTO 9780 
9340 ' 
9350 'PRINT FILENAME.EXT 
9360 PRINT#-2, G$(T1); 
9370 ' 
9380 'PRINT FILE TYPE 
9390 PRINT#-2, FT(T1); 
9400 ' 
9410 'PRINT THE GRANULES 
9420 'USED FROM FIRST TO 
9430 'LAST IN ORDER 
9440 PRINT#-2, H$(T1); 
9450 T3=1 'GRANULE COUNT 
9460 T2=FG(T1) 
9470 PRINT#-2, T2; 
9480 T2=FA(T2) 
9490 IF T2>67 GOTO 9570 
9500 T3=T3+1 
9510 PRINT#-2, T2; 
9520 GOTO 9480 
9530 ' 
9540 'PRINT THE NUMBER 
9550 'OF SECTORS USED IN 
9560 'THE LAST GRANULE 
9570 PRINT#-2, ">"; 
9580 T2=T2-&HC0 

9590 T$=STR$(T2) 
9600 L=LEN(T$) 
9610 T$=RIGHT$(T$, L-1) 
9620 PRINT#-2, T$; 
9630 ' 
9640 'PRINT THE NUMBER 
9650 'OF BYTES USED IN 
9660 'THE LAST SECTOR 
9670 PRINT#-2, " >>"; 
9680 T2=SB(T1) 
9690 T$=STR$(T2) 
9700 L=LEN(T$) 
9710 T$=RIGHT$(T$, L-1) 
9720 PRINT#-2, T$; 
9730 ' 
9740 'PRINT THE NUMBER 
9750 'OF GRANULES USED 
9760 PRINT#-2, T3 
9770 NEXT T1 
9780 PRINT#-2, " " 
9790 ' 
9800 'PRINT THE NUMBER 
9810 'OF FREE GRANULES 
9820 'ON THE DISK 
9830 PRINT#-2, "FREE = ";FREE(DN) 
9840 PRINT#-2, " " 
9850 ' 
9890 '***** 
9900 '* 
9910 '* CLOSE AND EXIT 
9920 '* 
9930 '***** 
9940 ' 
9950 CLS 
9960 PRINT@ 130, "DIRECTORY  PRINTOUT COMPLETE" 
9970 PRINT@ 388, "(PRESS ANY KEY TO EXIT)"  
9980 A$=INKEY$ 
9990 IF A$="" GOTO 9980 
32767 END 

 
===== 

Appendix A: New BDS Software License  
 

This New Software License applies to all software found on the BDS Software site, and 
supersedes all previous copyright notices and licensing provisions which may appear in the 
software itself or in any documentation therefor. 

 
All software which has previously been placed in the public domain remains in the public 

domain. 
 
All other software, programs, experiments and reports, documentation, and any other 

material on this site (other than that attributed to outside sources) is hereby copyright © 2018 (or 
later if so marked) by M. David Johnson. 

 
All software, documentation, and other information on the BDS Software site is available 

for you to freely download without cost. 
 
Whether you downloaded such items directly from this site, or you obtained them by any 

other means, you are hereby licensed to copy them, to sell or give away such copies, to use them, 
and to excerpt from them, in any way whatsoever, so long as nothing you do with them would 
denigrate the name of our Lord and Savior, Jesus Christ. 

 
I make absolutely no warranty whatsoever for any of these items. You use them entirely 

at your own risk. 
 
If they don't work for you, I commiserate. 
 
If they crash your system, I sympathize. 
 
But I accept no responsibility whatsoever for any such consequences. Under no 

circumstances will BDS Software or M. David Johnson be liable for any negative results of any 
kind which you may experience from downloading or using these items. 

 
BDS Software's former mail address at P.O. Box 485 in Glenview, IL is no longer valid. 

Any mail sent to that address will be rejected by the U.S. Postal Service. See my Contact page. 
 
M.D.J. 2018/06/08 
 
===== 

